

$Ax = b$, per alcuni valori di n , con il metodo diretto che utilizza EGPP e con il metodo iterativo di Jacobi. Per ciascun valore di n e ciascun metodo, si misurano: il *tempo di calcolo* (in secondi) e *l'errore assoluto* tra l'approssimazione e $x^* = \text{col}(1/4, 1/6, \dots, 1/6, 1/4)$, calcolato utilizzando la norma uno. Per il metodo iterativo si è scelto di eseguire 35 iterazioni (sufficienti per ottenere un'approssimazione con errore assoluto paragonabile a quello ottenuto con il metodo diretto) a partire da $g = 0$.

Risultati:

n	tempo di calcolo Jacobi	tempo di calcolo Gauss
100	3.0 d-2	1.2 d-1
500	1.3 d-1	8.2
1000	2.5 d-1	61
2000	5.1 d-1	troppa memoria per EGPP
3000	7.6 d-1	troppa memoria per EGPP
4000	1.0 (senza calcolare A)	troppa memoria per A
5000	1.3 (senza calcolare A)	troppa memoria per A

Per l'errore assoluto si ha, indipendentemente dal valore di n :

$$\text{err}(\text{Jacobi}) = 1.6 \text{ d-16} \quad , \quad \text{err}(\text{Gauss}) = 3.3 \text{ d-18}$$

Codice utilizzato:

```
//
clear;
//
Percorso = "...";
exec(Percorso + "EGPP.sci");
exec(Percorso + "SA.sci");
exec(Percorso + "SI.sci");
//
T_J = 0;
T_EG = 0;
NE = [];
//
N = 100;
ripetizioni = 50;
ITER = 35;
//
// commentare la riga seguente per n > 3000
A = 3*eye(N,N); A(2:N,1) = ones(N-1,1); A(1:N-1,N) = ones(N-1,1);
b = ones(N,1);
//
c = b/3;
//
x_esatta = ones(N,1)/6; x_esatta(1) = 1/4; x_esatta(N) = 1/4;
```

```

tic();
for j=1:ripetizioni,
    x = zeros(N,1);
    xn = x;
    for k = 1:ITER,
        s = -(x(1)+x(N))/3;
        for i = 2:N-1, xn(i) = s + c(i); end;
        xn(1) = c(1) - x(N)/3;
        xn(N) = c(N) - x(1)/3;
        x = xn;
        NE(k) = norm(x - x_esatta,1);
    end;
end;
T_J = toc()/ripetizioni;
//
printf("\nerrore relativo J = %4.3e\n",norm(x - x_esatta,1)/norm(x_esatta,1));
printf("\ntempo per Jacobi (N = %d) = %4.3e\n",N, T_J);
//
ERR_ZERO = find(NE == 0);
if ERR_ZERO ~= [] then NE = NE(1:min(ERR_ZERO)-1); end;
K = [1:length(NE)]';
scf(0);clf();
plot2d(K, [log10(3*NE(1)) - K * log10(3), log10(NE)], style = [2,5]);
xgrid();
xlabel('k (iterazione)'); legend('C' - k * log10(3)' , 'log10(err(k))');
//
tic();
//for j=1:ripetizioni,
    [S,D,P] = EGPP(A);
    [c] = SA(S,P*b);
    [x_EG] = SI(D,c);
//end;
T_EG = toc();
//
printf("\nerrore relativo EG = %4.3e\n",norm(x_EG - x_esatta,1)/norm(x_esatta,1));
printf("\ntempo per Gauss (N = %d)= %4.3e\n", N, T_EG);

```

- Il codice produce anche un grafico che consente di confrontare la successione generata dal metodo di Jacobi con la successione $C (1/3)^k$, con C opportuno, allo scopo di verificare la previsione teorica che $\| x(k) - x^* \| \rightarrow 0$ come $(1/3)^k$.


Il grafico riporta, in funzione dell'indice k, i valori di:

$$\log_{10}(\| x(k) - x^* \|) \quad \text{e} \quad \log_{10}(C (1/3)^k) = C' - k \log_{10}(3)$$

con $C' = \log(C)$ scelto in modo da rendere uguali i valori delle sequenze per $k = 1$:

$$C' = \log_{10}(3 \| x(1) - x^* \|)$$

Per $n = 5000$ si è ottenuto il grafico seguente (per gli altri valori di n si sono ottenuti grafici qualitativamente simili):



Si constata che la previsione teorica è fedelmente verificata almeno finché l'errore assoluto non assume valori molto piccoli.

(28.02) Osservazione (criteri d'arresto)

Siano A in $\mathbb{R}(n \times n)$ invertibile e b in $\mathbb{R}(n)$ non zero. Si utilizza il metodo iterativo convergente definito da H in $\mathbb{R}(n \times n)$ e c in $\mathbb{R}(n)$ per approssimare la soluzione x^* del sistema $A x = b$. Scelto g in $\mathbb{R}(n)$, il metodo iterativo genera la successione $\{ x(k) \}$, convergente ad x^* . Descriviamo due possibili *criteri d'arresto*.

(a) Assegnato $E > 0$ e posto $r(k) = b - A x(k)$ (vettore *residuo* associato ad $x(k)$):

$$\text{se } \| r(k) \| / \| b \| < E \text{ allora STOP}$$

- Il criterio è *calcolabile*;
- Il criterio è *efficace* (infatti: se $x(k) \rightarrow x^*$ allora $x(k) - x^* \rightarrow 0$ e quindi $r(k) = A (x^* - x(k)) \rightarrow 0$);
- Se il criterio è verificato si ha, interpretando $x(k)$ come soluzione del sistema perturbato $A x = b - r(k)$ ed utilizzando i risultati della teoria del condizionamento:

$$\| x(k) - x^* \| / \| x^* \| \leq c(A) \| r(k) \| / \| b \| < c(A) E$$

Il criterio risulta dunque di *tipo relativo*. Si osservi che se il numero di condizionamento di A è molto grande, ...

(b) Assegnato $E > 0$:

se $\|x(k) - x(k-1)\| < E$ allora STOP

- Il criterio è *calcolabile*;
- Il criterio è *efficace* (infatti: se $x(k) \rightarrow x^*$ allora $x(k-1) - x^* \rightarrow 0$ e quindi $x(k) - x(k-1) \rightarrow 0$);
- Se il criterio è verificato:

se $\|H\| < 1$ allora, posto $F(H) = \|H\| / (1 - \|H\|)$ si ha:

$$\|x(k) - x^*\| \leq F(H) \|x(k) - x(k-1)\| < F(H) E$$

Il criterio risulta dunque di *tipo assoluto*. Si osservi che se $\|H\|$ vale poco meno di uno allora $F(H)$ è molto grande e ...
